Productions highlighted in yellow are unchanged from 2396bis.

Productions highlighted in grey are unchanged from IRI.  
  absolute-xri    = "xri:" global-path 

  authority-path  = IRI-authority / XRI-authority

  dec-octet       = DIGIT                ; 0-9

                  / %x31-39 DIGIT        ; 10-99

                  / "1" 2DIGIT           ; 100-199

                  / "2" %x30-34 DIGIT    ; 200-249

                  / "25" %x30-35         ; 250-255
  gcs-char        = "+" / "=" / "@" / "$" / "*" / "!"
  gen-delims      = ":" / "/" / "?" / "#" / "[" / "]" / "@"
  global-path     = authority-path [ local-path ]

  h16             = 1*4HEXDIG
  iauthority      = [ iuserinfo "@" ] ihost [ ":" port ]

  ifragment       = *( ipchar / "/" / "?" )

  ihier-part      = "//" iauthority ipath-abempty

                  / ipath-abs

                  / ipath-rootless

                  / ipath-empty

  ihost           = IP-literal / IPv4address / ireg-name

  ipath-abempty   = *( "/" isegment )

  ipath-abs       = "/" [ isegment-nz *( "/" isegment ) ]

  ipath-empty     = 0<ipchar>

  ipath-noscheme  = isegment-nzc *( "/" isegment )

  ipath-rootless  = isegment-nz *( "/" isegment )

  ipchar          = iunreserved / pct-encoded / sub-delims / ":"

                  / "@"

  iprivate        = %xE000-F8FF / %xF0000-FFFFD / %x100000-10FFFD
  IP-literal      = "[" ( IPv6address / IPvFuture  ) "]"

  IPvFuture       = "v" 1*HEXDIG "." 1*( unreserved / sub-delims / ":" )

  IPv4address     = dec-octet "." dec-octet "." dec-octet "." dec-octet

  IPv6address     =                            6( h16 ":" ) ls32

                  /                       "::" 5( h16 ":" ) ls32

                  / [               h16 ] "::" 4( h16 ":" ) ls32

                  / [ *1( h16 ":" ) h16 ] "::" 3( h16 ":" ) ls32

                  / [ *2( h16 ":" ) h16 ] "::" 2( h16 ":" ) ls32

                  / [ *3( h16 ":" ) h16 ] "::"    h16 ":"   ls32

                  / [ *4( h16 ":" ) h16 ] "::"              ls32

                  / [ *5( h16 ":" ) h16 ] "::"              h16

                  / [ *6( h16 ":" ) h16 ] "::"
  iquery          = *( ipchar / iprivate / "/" / "?" )

  ireg-name       = 0*255( iunreserved / pct-encoded / sub-delims )

  IRI             = scheme ":" ihier-part [ "?" iquery ]

                  [ "#" ifragment ]
  IRI-authority   = "//" iauthority

  isegment        = *ipchar

  isegment-nz     = 1*ipchar

  isegment-nzc    = 1*( iunreserved / pct-encoded / sub-delims

                  / "@" )

  iunreserved     = ALPHA / DIGIT / "-" / "." / "_" / "~" / ucschar

  iuserinfo       = *( iunreserved / pct-encoded / sub-delims / ":" )
  local-path      = "/" relative-path

  ls32            = ( h16 ":" h16 ) / IPv4address

  pct-encoded     = "%" HEXDIG HEXDIG

  port            = *DIGIT
  relative-path   = [ xri-segments ] [ "?" xri-query ]

                  [ "#" xri-fragment ]

  relative-xri    = local-path / relative-path 

  reserved        = gen-delims / sub-delims

  sub-delims      = "!" / "$" / "&" / "'" / "(" / ")"

                  / "*" / "+" / "," / ";" / "="

  scheme          = ALPHA *( ALPHA / DIGIT / "+" / "-" / "." )
  sub-segment     = 1*xri-pchar / xref

  ucschar         = %xA0-D7FF / %xF900-FDCF / %xFDF0-FFEF /

                  %x10000-1FFFD / %x20000-2FFFD / %x30000-3FFFD /

                  %x40000-4FFFD / %x50000-5FFFD / %x60000-6FFFD /

                  %x70000-7FFFD / %x80000-8FFFD / %x90000-9FFFD /

                  %xA0000-AFFFD / %xB0000-BFFFD / %xC0000-CFFFD /

                  %xD0000-DFFFD / %xE1000-EFFFD
  unreserved      = ALPHA / DIGIT / "-" / "." / "_" / "~"
  xref            = "(" ( xri-value / IRI ) ")"

  xref-authority  = xref  *( "*" sub-segment / ":" sub-segment )

  XRI             = absolute-xri / relative-xri

  XRI-authority   = ( gcs-char [ xri-segment ] ) / xref-authority

  xri-characters  = xri-reserved / xri-unreserved / pct-encoded
  xri-fragment    = [ xref ] * ( xri-pchar / "/" / "?"
                  /  ":" / "/" / "?" / "[" / "]" / "@" / "(" / ")"
                  / "!" / "$" / "*" / "+" / "=" )
  xri-gen-delims  = ":" / "/" / "?" / "#" / "[" / "]" / "@" / "("

                  / ")" / "!" / "$" / "*" / "+" / "="

  xri-pchar       = xri-unreserved / pct-encoded / xri-sub-delims

  xri-query       = [ xref ] *( xri-pchar / iprivate / "/" / "?"
                  /  ":" / "/" / "?" / "[" / "]" / "@" / "(" / ")"

                  / "!" / "$" / "*" / "+" / "=" )

  xri-reserved    = xri-gen-delims / xri-sub-delims

  xri-segment     = ( [ "*" ] sub-segment ) / ( ":" sub-segment ) 

                  *( "*" sub-segment / ":" sub-segment )
  xri-segments    = xri-segment *( "/" [ xri-segment ] )

  xri-sub-delims  = "&" / "'" / "," / ";" 

  xri-unreserved  = unreserved / ucschar

  xri-value       = global-path / local-path / relative-path
Issues

1) We allow fragment in an absolute XRI. This is disallowed in URI and IRI.
2) xri-segments (and thus relative-path) can be foo/////bar. OK? Note that this is also possible in conventional URIs under 2396bis.
3) Why is xref mentioned explicitly in xri-query? I’d rather remove it an just allow parens. If we don’t do this, we’ll require percent encoding of parens in XRIs included in a query’s name/value pair. Same comment for fragment.

4) All xri-gen-delims are disallowed in a path, so they must be percent encoded if they don’t have their defined syntactic meaning. For example, something like xri:@foo/@bar is illegal - the second @ sign must be percent encoded. OK?

